
clear all

use Master2018.dta

// The data is next collapsed at the Lok Sabha overall session level

replace LS_overall = LS_overall[_n-1] if missing(LS_overall)
#delimit ;
collapse
(first) 
LSmaj RSmaj LegisMaj onlyLSmaj onlyRSmaj prime_minister PMmandate govt_type bicam_type
totalloksabha mainpartygovtseats mainoppositionseats coalitionpartners
date longdate dayofweek day2 dateinmonth month month_id year yr_bef_elec
noofbillspassedbybothhousesofpar totalbillsthatwerepreviousordina
loksabhainsession sessionwithinloksabha effectpartseat effectpartvote cong_govt frac_slp

(sum)
totalordinances = ordinances
session_length = LSsession
break_length = LSbreak
repromulgation
became_bill
, by(LS_overall);
#delimit cr

label var noofbillspassedbybothhousesofpar "Number of Acts"
label var break_length "Length of Break"
label var PMmandate "PM had mandate"
label var LegisMaj "Bicameral majority"
label var LSmaj "LS Majority"

tsset LS_overall
replace totalbillsthatwerepreviousordina = l.repromulgation if date > 39503


//The next two lines correct the data errors in totalbillsthatwerepreviousordina
replace totalbillsthatwerepreviousordina = 1 if date == 31831
replace totalbillsthatwerepreviousordina = l.totalordinances if totalbillsthatwerepreviousordina >  l.totalordinances

program drregx, eclass
	version 14
	syntax varlist(min=2 fv) [if] [in], robust [Test(string)] [*]
	reg `varlist' `if' `in', robust `options'
	if "`test'" != "" {
		test `test'
		ereturn local test_cmd `test'
		ereturn scalar test_p = r(p)
	}
end

rename date date1

gen date = (date1 - 31910) / 3650

gen post_1985 = 0
replace post_1985 = 1 if date1 >=31776 //Wadhwa decision 20th Dec 1986


program regx, eclass
	version 14
	syntax varlist(min=2 fv) [if] [in], robust [Test(string)] [TEST2(string)] [TEST3(string)] [TEST4(string)] [YMEAN(string)] [*]
	reg `varlist' `if' `in', robust `options'
	if "`test'" != "" {
		test `test'
		ereturn local test_cmd `test'
		ereturn scalar test_p = r(p)
	}
	if "`test2'" != "" {
		test `test2'
		ereturn local test2_cmd `test2'
		ereturn scalar test2_p = r(p)
	}
	if "`test3'" != "" {
		test `test3'
		ereturn local test3_cmd `test3'
		ereturn scalar test3_p = r(p)
	}
end

label drop govt_type

#delimit ;
label define govt_type
0 "Single Party Majority"
1 "Single Party Minority" 
2 "Majority Party Coalition" 
3 "Minority Coalition" 
;
#delimit cr

label values govt_type govt_type

label drop bicam_type

#delimit ;
label define bicam_type
0 "Bicameral Majority"
1 "Minority in Both" 
2 "Only Minority in RS" 
3 "Only Minority in LS"
;
#delimit cr

label values bicam_type bicam_type

label var post_1985 "Post 1986"

label var yr_bef_elec "Yr before  election"
label var effectpartvote "Effective num parties"
label var cong_govt "Congress govt"
label var date "Linear time trend"



egen meanbills = mean(noofbillspassedbybothhousesofpar)
gen totalbills = 	noofbillspassedbybothhousesofpar - meanbills

gen unconverted_ords = l.totalordinances - totalbillsthatwerepreviousordina 
//For the unconverted ords drop the 1st session of LS:
replace unconverted_ords =. if loksabhainsession == 1 + l.loksabhainsession

gen prop_unconverted_ords = unconverted_ords / (l.totalordinances + .01)
//Adding a small positive constant .01 to get around the 0 ordinance breaks

label var totalbills "Number of Acts"

local session_ctls "break_length session_length i.sessionwithinloksabha"
local pol_ctls "yr_bef_elec effectpartvote cong_govt"

// The next 6 lines replicate the regressions in Table 5 from the paper
eststo substitution_1: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo substitution_2: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj  i.month_id, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo substitution_3: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj `session_ctls'  i.month_id, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo substitution_4: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj `session_ctls'  `pol_ctls' i.month_id, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo substitution_5: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj 		  `session_ctls'  yr_bef_elec i.month_id i.loksabhainsession i.month_id, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo substitution_6: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj 		  `session_ctls'  yr_bef_elec i.month_id i.year, robust test(totalbills + totalbills#c.LegisMaj = 0)


eststo LS_substitution_1: drregx totalordinances LSmaj totalbills c.totalbills#c.LSmaj, robust test(totalbills + totalbills#c.LSmaj = 0)
eststo LS_substitution_2: drregx totalordinances LSmaj totalbills c.totalbills#c.LSmaj  i.month_id, robust test(totalbills + totalbills#c.LSmaj = 0)
eststo LS_substitution_3: drregx totalordinances LSmaj totalbills c.totalbills#c.LSmaj `session_ctls'  i.month_id, robust test(totalbills + totalbills#c.LSmaj = 0)
eststo LS_substitution_4: drregx totalordinances LSmaj totalbills c.totalbills#c.LSmaj `session_ctls'  `pol_ctls' i.month_id, robust test(totalbills + totalbills#c.LSmaj = 0)
eststo LS_substitution_5: drregx totalordinances LSmaj totalbills c.totalbills#c.LSmaj 	   `session_ctls'  yr_bef_elec i.month_id i.loksabhainsession i.month_id, robust test(totalbills + totalbills#c.LSmaj = 0)
eststo LS_substitution_6: drregx totalordinances LSmaj totalbills c.totalbills#c.LSmaj 	   `session_ctls'  yr_bef_elec i.month_id i.year, robust test(totalbills + totalbills#c.LSmaj = 0)


// The next 5 lines replicate the regressions in Table 3 from the paper
eststo morebills_1: reg noofbillspassedbybothhousesofpar LegisMaj, robust
eststo morebills_2: reg noofbillspassedbybothhousesofpar LegisMaj i.month_id, robust
eststo morebills_3: reg noofbillspassedbybothhousesofpar LegisMaj `session_ctls'  i.month_id, robust
eststo morebills_4: reg noofbillspassedbybothhousesofpar LegisMaj `session_ctls' `pol_ctls' i.month_id, robust
eststo morebills_5: reg noofbillspassedbybothhousesofpar LegisMaj `session_ctls' `pol_ctls' i.month_id date, robust


// eststo TS_morebills_1: reg noofbillspassedbybothhousesofpar LegisMaj post_1985 c.LegisMaj#c.post_1985 , robust
// eststo TS_morebills_2: reg noofbillspassedbybothhousesofpar LegisMaj post_1985 c.LegisMaj#c.post_1985 i.month_id, robust
// eststo TS_morebills_3: reg noofbillspassedbybothhousesofpar LegisMaj post_1985 c.LegisMaj#c.post_1985 `session_ctls'  i.month_id, robust
// eststo TS_morebills_4: reg noofbillspassedbybothhousesofpar LegisMaj post_1985 c.LegisMaj#c.post_1985 `session_ctls' `pol_ctls' i.month_id, robust
// eststo TS_morebills_5: reg noofbillspassedbybothhousesofpar LegisMaj post_1985 c.LegisMaj#c.post_1985 `session_ctls' `pol_ctls' i.month_id date, robust


eststo TS_pre_subs_1: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj if post_1985 == 0, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pre_subs_2: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj  i.month_id if post_1985 == 0, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pre_subs_3: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj `session_ctls'  i.month_id if post_1985 == 0, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pre_subs_4: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj `session_ctls'  `pol_ctls' i.month_id if post_1985 == 0, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pre_subs_5: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj 		  `session_ctls'  yr_bef_elec i.month_id i.loksabhainsession i.month_id if post_1985 == 0, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pre_subs_6: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj 		  `session_ctls'  yr_bef_elec i.month_id i.year if post_1985 == 0, robust test(totalbills + totalbills#c.LegisMaj = 0)

eststo TS_pst_subs_1: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj if post_1985 == 1, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pst_subs_2: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj  i.month_id if post_1985 == 1, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pst_subs_3: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj `session_ctls'  i.month_id if post_1985 == 1, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pst_subs_4: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj `session_ctls'  `pol_ctls' i.month_id if post_1985 == 1, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pst_subs_5: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj 		  `session_ctls'  yr_bef_elec i.month_id i.loksabhainsession i.month_id if post_1985 == 1, robust test(totalbills + totalbills#c.LegisMaj = 0)
eststo TS_pst_subs_6: drregx totalordinances LegisMaj totalbills c.totalbills#c.LegisMaj 		  `session_ctls'  yr_bef_elec i.month_id i.year if post_1985 == 1, robust test(totalbills + totalbills#c.LegisMaj = 0)


eststo LS_morebills_1: reg noofbillspassedbybothhousesofpar LSmaj, robust
eststo LS_morebills_2: reg noofbillspassedbybothhousesofpar LSmaj i.month_id, robust
eststo LS_morebills_3: reg noofbillspassedbybothhousesofpar LSmaj `session_ctls' i.month_id, robust
eststo LS_morebills_4: reg noofbillspassedbybothhousesofpar LSmaj `session_ctls' `pol_ctls' i.month_id, robust
eststo LS_morebills_5: reg noofbillspassedbybothhousesofpar LSmaj `session_ctls' `pol_ctls' i.month_id date, robust


eststo type_subs_1: regx totalordinances i.govt_type totalbills c.totalbills#(1.govt_type 2.govt_type 3.govt_type), robust test(totalbills#1.govt_type = totalbills#2.govt_type) test2(totalbills#1.govt_type = totalbills#3.govt_type) test3(totalbills#2.govt_type = totalbills#3.govt_type)
eststo type_subs_2: regx totalordinances i.govt_type totalbills c.totalbills#(1.govt_type 2.govt_type 3.govt_type)  i.month_id, robust test(totalbills#1.govt_type = totalbills#2.govt_type) test2(totalbills#1.govt_type = totalbills#3.govt_type) test3(totalbills#2.govt_type = totalbills#3.govt_type)
eststo type_subs_3: regx totalordinances i.govt_type totalbills c.totalbills#(1.govt_type 2.govt_type 3.govt_type) `session_ctls'  i.month_id, robust test(totalbills#1.govt_type = totalbills#2.govt_type) test2(totalbills#1.govt_type = totalbills#3.govt_type) test3(totalbills#2.govt_type = totalbills#3.govt_type)
eststo type_subs_4: regx totalordinances i.govt_type totalbills c.totalbills#(1.govt_type 2.govt_type 3.govt_type) `session_ctls'  `pol_ctls' i.month_id, robust test(totalbills#1.govt_type = totalbills#2.govt_type) test2(totalbills#1.govt_type = totalbills#3.govt_type) test3(totalbills#2.govt_type = totalbills#3.govt_type)
eststo type_subs_5: regx totalordinances i.govt_type totalbills c.totalbills#(1.govt_type 2.govt_type 3.govt_type) `session_ctls'  yr_bef_elec i.month_id i.loksabhainsession i.month_id, robust test(totalbills#1.govt_type = totalbills#2.govt_type) test2(totalbills#1.govt_type = totalbills#3.govt_type) test3(totalbills#2.govt_type = totalbills#3.govt_type)
eststo type_subs_6: regx totalordinances i.govt_type totalbills c.totalbills#(1.govt_type 2.govt_type 3.govt_type) `session_ctls'  yr_bef_elec i.month_id i.year, robust test(totalbills#1.govt_type = totalbills#2.govt_type) test2(totalbills#1.govt_type = totalbills#3.govt_type) test3(totalbills#2.govt_type = totalbills#3.govt_type)


eststo type_morebills_1: regx noofbillspassedbybothhousesofpar i.govt_type, robust test(1.govt_type = 2.govt_type) test2(1.govt_type = 3.govt_type) test3(2.govt_type = 3.govt_type)
eststo type_morebills_2: regx noofbillspassedbybothhousesofpar i.govt_type i.month_id, robust test(1.govt_type = 2.govt_type) test2(1.govt_type = 3.govt_type) test3(2.govt_type = 3.govt_type)
eststo type_morebills_3: regx noofbillspassedbybothhousesofpar i.govt_type i.month_id `session_ctls', robust test(1.govt_type = 2.govt_type) test2(1.govt_type = 3.govt_type) test3(2.govt_type = 3.govt_type)
eststo type_morebills_4: regx noofbillspassedbybothhousesofpar i.govt_type i.month_id `session_ctls' `pol_ctls' , robust test(1.govt_type = 2.govt_type) test2(1.govt_type = 3.govt_type) test3(2.govt_type = 3.govt_type)
eststo type_morebills_5: regx noofbillspassedbybothhousesofpar i.govt_type i.month_id `session_ctls' `pol_ctls' date, robust test(1.govt_type = 2.govt_type) test2(1.govt_type = 3.govt_type) test3(2.govt_type = 3.govt_type)


eststo TS_pre_bills_1: reg noofbillspassedbybothhousesofpar LegisMaj if post_1985 == 0, robust
eststo TS_pre_bills_2: reg noofbillspassedbybothhousesofpar LegisMaj i.month_id if post_1985 == 0, robust
eststo TS_pre_bills_3: reg noofbillspassedbybothhousesofpar LegisMaj `session_ctls'  i.month_id if post_1985 == 0, robust
eststo TS_pre_bills_4: reg noofbillspassedbybothhousesofpar LegisMaj `session_ctls' `pol_ctls' i.month_id if post_1985 == 0, robust
eststo TS_pre_bills_5: reg noofbillspassedbybothhousesofpar LegisMaj `session_ctls' `pol_ctls' i.month_id date if post_1985 == 0, robust

eststo TS_pst_bills_1: reg noofbillspassedbybothhousesofpar LegisMaj if post_1985 == 1, robust
eststo TS_pst_bills_2: reg noofbillspassedbybothhousesofpar LegisMaj i.month_id if post_1985 == 1, robust
eststo TS_pst_bills_3: reg noofbillspassedbybothhousesofpar LegisMaj `session_ctls'  i.month_id if post_1985 == 1, robust
eststo TS_pst_bills_4: reg noofbillspassedbybothhousesofpar LegisMaj `session_ctls' `pol_ctls' i.month_id if post_1985 == 1, robust
eststo TS_pst_bills_5: reg noofbillspassedbybothhousesofpar LegisMaj `session_ctls' `pol_ctls' i.month_id date if post_1985 == 1, robust




eststo bicam_type_subs_1: regx totalordinances i.bicam_type totalbills c.totalbills#(1.bicam_type 2.bicam_type), robust test(totalbills#1.bicam_type = totalbills#2.bicam_type)
eststo bicam_type_subs_2: regx totalordinances i.bicam_type totalbills c.totalbills#(1.bicam_type 2.bicam_type)  i.month_id, robust test(totalbills#1.bicam_type = totalbills#2.bicam_type)
eststo bicam_type_subs_3: regx totalordinances i.bicam_type totalbills c.totalbills#(1.bicam_type 2.bicam_type) `session_ctls'  i.month_id, robust test(totalbills#1.bicam_type = totalbills#2.bicam_type)
eststo bicam_type_subs_4: regx totalordinances i.bicam_type totalbills c.totalbills#(1.bicam_type 2.bicam_type) `session_ctls'  `pol_ctls' i.month_id, robust test(totalbills#1.bicam_type = totalbills#2.bicam_type)
eststo bicam_type_subs_5: regx totalordinances i.bicam_type totalbills c.totalbills#(1.bicam_type 2.bicam_type) `session_ctls'  yr_bef_elec i.month_id i.loksabhainsession i.month_id, robust test(totalbills#1.bicam_type = totalbills#2.bicam_type)
eststo bicam_type_subs_6: regx totalordinances i.bicam_type totalbills c.totalbills#(1.bicam_type 2.bicam_type) `session_ctls'  yr_bef_elec i.month_id i.year, robust test(totalbills#1.bicam_type = totalbills#2.bicam_type)



eststo bicam_type_morebills_1: regx noofbillspassedbybothhousesofpar i.bicam_type, robust test(1.bicam_type = 2.bicam_type)
eststo bicam_type_morebills_2: regx noofbillspassedbybothhousesofpar i.bicam_type i.month_id, robust test(1.bicam_type = 2.bicam_type)
eststo bicam_type_morebills_3: regx noofbillspassedbybothhousesofpar i.bicam_type i.month_id `session_ctls', robust test(1.bicam_type = 2.bicam_type)
eststo bicam_type_morebills_4: regx noofbillspassedbybothhousesofpar i.bicam_type i.month_id `session_ctls' `pol_ctls' , robust test(1.bicam_type = 2.bicam_type)
eststo bicam_type_morebills_5: regx noofbillspassedbybothhousesofpar i.bicam_type i.month_id `session_ctls' `pol_ctls' date, robust test(1.bicam_type = 2.bicam_type)




#delimit ;
local common_tab_opts
star(* 0.1 ** 0.05 *** 0.01)
varwidth(30) modelwidth(10) wrap se r2 nobase noomitted 
label numbers nodepvars
;
#delimit cr

local all_tab_opts		`common_tab_opts' indicate("Month dummies = *.month_id" "Session controls = break_length session_length *.sessionwithinloksabha" )
local morebills_opts	`common_tab_opts' indicate("Month dummies = *.month_id" "Session controls = break_length session_length *.sessionwithinloksabha" )
local year_and_ls		`common_tab_opts' indicate("Month dummies = *.month_id" "Session controls = break_length session_length *.sessionwithinloksabha" "Lok Sabha dummies = *.loksabhainsession" "Year dummies = *.year")


esttab substitution_* 	 , `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0: \beta_A + \lambda = 0\$" ) sfmt(3)
esttab LS_substitution_* , `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0: \beta_A + \lambda = 0\$" ) sfmt(3)
esttab morebills_* 		 , `morebills_opts' nomtitles
esttab LS_morebills_* 	 , `morebills_opts' nomtitles
esttab type_morebills_*  , `morebills_opts' nomtitles scalar("test_p \$p\$-value \$H_0^1: \gamma_1 = \gamma_2\$" "test2_p \$p\$-value \$H_0^2: \gamma_1 = \gamma_3\$"  "test3_p \$p\$-value \$H_0^3: \gamma_2 = \gamma_3\$" ) sfmt(3)
esttab type_subs_* 	     , `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0^1: \gamma_1 = \gamma_2\$" "test2_p \$p\$-value \$H_0^2: \gamma_1 = \gamma_3\$"  "test3_p \$p\$-value \$H_0^3: \gamma_2 = \gamma_3\$" ) sfmt(3)


esttab TS_pre_bills* 		 , `morebills_opts' nomtitles
esttab TS_pst_bills* 		 , `morebills_opts' nomtitles

esttab TS_pre_subs* 	 , `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0: \beta_A + \lambda = 0\$" ) sfmt(3)
esttab TS_pst_subs* 	 , `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0: \beta_A + \lambda = 0\$" ) sfmt(3)
esttab bicam_type_morebills_*  , `morebills_opts' nomtitles scalar("test_p \$p\$-value \$H_0^1: \gamma_1 = \gamma_2\$") sfmt(3)
esttab bicam_type_subs_* 	     , `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0^1: \gamma_1 = \gamma_2\$") sfmt(3)
//esttab TS_morebills_* 		 , `morebills_opts' nomtitles


//Latex


esttab substitution_*	 using "tables/substitution.tex"  	, `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0: \beta_A + \lambda = 0\$" ) substitute(\_ _) sfmt(3) replace
esttab LS_substitution_* using "tables/LS_substitution.tex" , `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0: \beta_A + \lambda = 0\$" ) substitute(\_ _) sfmt(3) replace
esttab morebills_* 		 using "tables/morebills.tex"  		, `morebills_opts' nomtitles replace
esttab LS_morebills_* 	 using "tables/LS_morebills.tex"  	, `morebills_opts' nomtitles replace
esttab type_subs_* 	     using "tables/type_subs.tex" 		, `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0^1: \gamma_1 = \gamma_2\$" "test2_p \$p\$-value \$H_0^2: \gamma_1 = \gamma_3\$"  "test3_p \$p\$-value \$H_0^3: \gamma_2 = \gamma_3\$" ) sfmt(3) substitute(\_ _) replace
esttab type_morebills_*  using "tables/type_morebills.tex"  , `morebills_opts' nomtitles scalar("test_p \$p\$-value \$H_0^1: \gamma_1 = \gamma_2\$" "test2_p \$p\$-value \$H_0^2: \gamma_1 = \gamma_3\$"  "test3_p \$p\$-value \$H_0^3: \gamma_2 = \gamma_3\$" ) sfmt(3) substitute(\_ _) replace
esttab bicam_type_subs_* using "tables/bicam_type_subs.tex" , `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0^1: \gamma_1 = \gamma_2\$") sfmt(3) substitute(\_ _) replace
esttab bicam_type_morebills_*  using "tables/bicam_type_morebills.tex"  , `morebills_opts' nomtitles scalar("test_p \$p\$-value \$H_0^1: \gamma_1 = \gamma_2\$") sfmt(3) substitute(\_ _) replace
//esttab TS_morebills_* 	 using "tables/TS_morebills.tex" 	, `morebills_opts' nomtitles replace
esttab TS_pre_subs*	 	using "tables/TS_pre_subs.tex"  	, `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0: \beta_A + \lambda = 0\$" ) substitute(\_ _) sfmt(3) replace
esttab TS_pst_subs*	 	using "tables/TS_pst_subs.tex"  	, `year_and_ls' nomtitles scalar("test_p \$p\$-value \$H_0: \beta_A + \lambda = 0\$" ) substitute(\_ _) sfmt(3) replace
esttab TS_pre_bills* 	using "tables/TS_pre_bills.tex"  	, `morebills_opts' nomtitles replace
esttab TS_pst_bills* 	using "tables/TS_pst_bills.tex"  	, `morebills_opts' nomtitles replace
